

This file contains two macros for use in Microsoft Word for Windows which permit you to do automatic faxing from Microsoft Word. These macros are intended as examples of how DDE can be used from other programs to help automate faxing with Eclipse FAX. Consult your Word documentation for detailed instructions on using mail merge and macros within MS Word.

BulkMail

The following is a mail merge macro that will fax your merged documents using E-FAX's DDE capability. You will find that it operates without much user intervention; however, there are a few things you must know before you begin.

Note:

For information about how to use mail merge in MS-Word, or for information on how to install and use a macro in Word, follow the instructions in the *Microsoft Word User's Guide*.

The following are some specific points to be aware of:

1. This macro searches for a fax number imbedded in your merged document. In order for the macro to find it, you must place the merge field for the fax number in your document like this:

«fax» ~fax

The merge field for the fax number must be at the beginning of a line, and the word '~fax' (including the tilde) must appear one space following the merge field. You may have other text following it, and you may place it anywhere in the document. If you don't want the number to show on the faxed document, you can format the fax number field as 'hidden text'.

2. Currently, E-FAX won't perform the standard phonebook operations on phone numbers passed to it using DDE, so the phone numbers in your data table **MUST** have the proper area code (preceded by a 1) and any dialing prefix your office may require. We hope to have this feature added in a future release of E-FAX.
3. After creating your form letter and your data table, attach the data file to the form letter. Next, merge the two together by choosing the option "Merge to New Document"; this will create a new Word file with all of your merged letters. For more specific and complete instructions on creating and using form letters and data files, refer to your Microsoft Word documentation.
4. If E-FAX is not running when the macro is executed, it will be launched. The macro assumes the program is in the drive\directory C:\EFAX. If your program is located in another drive or directory, be sure to change the line in the macro that contains the path to the program (this is the line that reads "Shell "C:\EFAX\E-FAXMGR.EXE -nologo", 0).
5. If you have pending faxes in the queue, E-FAX will accept all of the new documents from the merge file but will not begin sending until you dismiss the dialog asking if you wish to view the send log.

```

Declare Function IsAppLoaded Lib "kernel"(name$) As Integer Alias "GetModuleHandle"
Dim Shared HoldPrinter$
Dim Shared FaxNumber$
Dim Shared NumSections

Sub MAIN
  CheckEnv
  SectionCount
  While NumSections
    GetFaxNumber
    DoDDE
    MarkOneDoc
    DoPrint
    NumSections = NumSections - 1
  Wend
  RestoreEnv
End Sub

Sub SectionCount
  EndOfDocument
  NumSections = SelInfo(2) - 1
End Sub

Sub MarkOneDoc
  StartOfDocument
  EditGoTo .Destination = "s"
  LineUp
  StartOfDocument 1
End Sub

Sub DoPrint
  FilePrint .Type = 0, .NumCopies = "1", .Range = 1, .From = "", .To = "", .PrintToFile = 0, \
  .Collate = 1, .FileName = ""
  EditClear
  EditClear
End Sub

Sub GetFaxNumber
  StartOfDocument
  EditFind .Find = "~fax", .WholeWord = 0, .MatchCase = 0, .Direction = 1, .Format = 0
  CharLeft 2
  StartOfLine 1
  FaxNumber$ = Selection$()
End Sub

Sub DoDDE
  ChanNumber = DDEInitiate("Eclipse FAX Manager", "SEND")
  DDEPoke(ChanNumber, "SEND", FaxNumber$)
  DDETerminate(ChanNumber)
End Sub

Sub CheckEnv
  HoldPrinter$ = GetProfileString$("windows", "device")
  IsInstalled = InStr(GetProfileString$("devices", "E-FAX"), "E-FAX")
  If IsInstalled Then
    SetPrinter
  Else
    MsgBox "Eclipse FAX is not installed. Please install Eclipse FAX before using this macro.", "BulkMail", 16
  End If
End Sub

```

```
        Stop
    End If
    ViewNormal
    ToolsOptionsView .Hidden = 1
    If IsAppLoaded("E-FAXMGR") = 0 Then
        Shell "C:\EFAX\E-FAXMGR.EXE -nologo -nopending", 0
    End If
End Sub

Sub SetPrinter
    Cnt = Len(GetProfileString$("devices", "E-FAX"))
    If Cnt = 11 Then
        FilePrintSetup .Printer = "E-FAX on " + Right$(GetProfileString$("devices", "E-FAX"), 5)
    Else
        FilePrintSetup .Printer = "E-FAX on CAS"
    End If
End Sub

Sub RestoreEnv
    ToolsOptionsView .Hidden = 0
    SetProfileString "windows", "device", HoldPrinter$
End Sub
```

SendFax

The README.WRI file includes a macro for use in MS-Word for Windows, Version 2.0. This macro uses a name or fax number in your document to send the fax. Before you can use the Word macro, you must have done the following:

- Installed the MS-Word macro
- Entered the relevant phone number in the Eclipse FAX phonebook
- Selected the correct Eclipse FAX phonebook

Note:

For information on how to install and use a macro in Word, follow the instructions in the *Microsoft Word User's Guide*.

1. In your Word document, highlight the name of the person to whom you want to fax the document.
2. Run the Send Fax macro. The Microsoft Word User's Guide tells how to run macros.

The macro selects the correct printer to send out a fax, looks up the recipient's name in the phonebook, dials the fax number, and transmits the fax. The DDE function only searches the currently selected phonebook for the highlighted name. If the name is not in that phonebook, or if the name is a duplicate, the Send Fax dialog box appears. From there you can add the name or look up the proper spelling. After sending the fax, the macro resets your system to the default printer. Another way to use this macro is to highlight a fax number instead of a name. Then, when you run the macro, it simply dials the highlighted number and transmits the fax.

```

Declare Function IsAppLoaded Lib "kernel"(name$) As Integer Alias "GetModuleHandle"
Dim Shared HoldPrinter$
Dim Shared FaxNumber$

Sub MAIN
  CheckEnv
  FaxNumber$ = Selection$()
  DoDDE
  FilePrintDefault
  RestoreEnv
End Sub

Sub CheckEnv
  HoldPrinter$ = GetProfileString$("windows", "device")
  IsInstalled = InStr(GetProfileString$("devices", "E-FAX"), "E-FAX")
  If IsInstalled Then
    SetPrinter
  Else
    MsgBox "Eclipse FAX is not installed. Please install Eclipse FAX before using this macro.", "SendFax", 16
    Stop
  End If
  ViewNormal
  ToolsOptionsView .Hidden = 1
  If IsAppLoaded("E-FAXMGR") = 0 Then
    Shell "C:\WINDOWS\EFAX\E-FAXMGR.EXE -nologo -nopending", 0
  End If
End Sub

Sub SetPrinter
  Cnt = Len(GetProfileString$("devices", "E-FAX"))
  If Cnt = 11 Then
    FilePrintSetup .Printer = "E-FAX on " + Right$(GetProfileString$("devices", "E-FAX"), 5)
  Else
    FilePrintSetup .Printer = "E-FAX on CAS"
  End If
End Sub

Sub DoDDE
  ChanNumber = DDEInitiate("Eclipse FAX Manager", "SEND")
  DDEPoke(ChanNumber, "SEND", FaxNumber$)
  DDETerminate(ChanNumber)
End Sub

Sub RestoreEnv
  ToolsOptionsView .Hidden = 0
  SetProfileString "windows", "device", HoldPrinter$
End Sub

```

Once Eclipse FAX has passed the name (or phone number) from the other Windows application and is activated through the Print Command it will attempt to look the name up in the current phonebook. If it finds the name it will send the fax out using the settings last used in the Send Fax dialog. If E-FAX does not find the name, or there is a duplicate name, the Send Fax dialog will appear as usual. The macro is case sensitive; if you do not capitalize the name in your Word document Eclipse FAX will not find the name and the Send Dialog will be presented. Also, if you highlight a carriage return at the end of the name, Eclipse FAX will not match the name with the entry in the phonebook.

Note:

If you highlight a fax number instead of a name the document will be faxed using the number you have highlighted, regardless of whether or not the person is in the phonebook.